

SAP Community > Products and Technology > Enterprise Resource Planning  
> ERP Blogs by Members > How To Add Batch Verification To WM Mobil Screen W...

## How To Add Batch Verification To WM Mobil Screen With Enter Definition



**beyhan\_meyrali**  
Active Contributor

2022 Aug 30 3:23 PM



7 Kudos

2,397 Views


Hi,

I don't know why but batch verification does not exists on WM mobile screens. I have checked Google and noticed many people are asking same question. Therefore I decided to share my solution with you.

If you worked with WM, you would know, WM provides RF Screens to be used by mobile devices with tcodes starts with LMXX. Those screens can be enhanced and new screen can be assigned to those transaction codes with a variant. And in LRFMD you can choose that variant if you want to work with enhanced screen.

Replacing screens and adding new fields are easy. For that I have followed Rakshith Gore's [blog.post](#). That part is easy. It becomes complex once you need to integrate Verification customization to that new screen. And on top of that, if you need to implement Enter Functionality customization, then you may spend sometime debugging existing screens.

Please check screen shot below to see [customizing settings](#) for Verification and Enter Behavior.




## Verification Profile Settings

Table View Edit Goto Selection Utilities System Help

Change View **Definition of Verification Profile**: Overview of Selected

New Entries No Batch verification setting here :/


WhN	Profile	Source bin	Dest.bin	SU src. bin	SU Dest. bin	Qty src. bin	Qty dest. bin	Mat.src.t
TST	INTMOV	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	1
TST	PCKEAC	<input checked="" type="checkbox"/>	<input type="checkbox"/>	<input checked="" type="checkbox"/>	<input type="checkbox"/>	<input checked="" type="checkbox"/>	<input type="checkbox"/>	1
TST	PICK	<input checked="" type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	1
TST	PUTWY	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input checked="" type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	1




## Enter Function Settings

Remember that part after copying screen 1212 to 9212. Here you need to write 1212 not 9212!

Note: 1212 and 9212 screens are explained below.





## Enhancement bit...

So far we have seen customization settings and below, I will show you how to add batch verification to LM06 screen.

First we need to find the screen and enhancement for it. To do that, we just need to open status screen on LM06 screen that we want to change.

The screenshot shows the 'System: Status' dialog box in SAP. The 'Source information' section on the left includes fields for Source Bin (SL1 01-01-02), Material (50), SSgt (1), and TL\_WHM (2000000152). A red box highlights the 'New Field' button next to the TL\_WHM field. The 'SAP data' section on the right shows repository data with Transaction (LM05), Program (screen) (SAPLLMOB), and Program (GUI) (SAPLLMOB). The Screen number is 2212, and the GUI status is LMOB\_GROUP3. A red arrow points to the '2212' field with the text 'LAST 3 Digits!'.

Our screen number is 2212. Last 3 digit is important, \*212 will help us to find user exit.

To find user exit, run tcode SMOD and select components, enter \*wm\*212\* and pres f4 to search.

The left screenshot shows the 'SAP Enhancements' screen with the search criteria '\*wm\*212\*' entered in the 'Enhancement' field. A red arrow points to the search button. The 'Object Components' section has 'Components' selected. The right screenshot shows the same screen with the search criteria 'MWMRF212' entered in the 'Enhancement' field. A red arrow points to the search button. The 'Object Components' section has 'Components' selected.

We found our user-exit. Now click display and pay attention to program name.

The screenshot shows the SAP Enhancement MWMRF212 interface. It has a menu bar with 'Enhancement', 'Edit', 'Goto', 'Environment', 'System', and 'Help'. Below the menu is a toolbar with various icons. The main content area is titled 'Components in SAP Enhancement MWMRF212' and contains three sections:

- Function Exits:** A table with two columns: 'Function module' and 'Short Text'. It lists two entries: 'EXIT\_SAPLLMOB\_170' with short text 'USER EXIT - DATA TO CUSTOMER SCREEN (DUMMY 1212)' and 'EXIT\_SAPLLMOB\_480' with short text 'USER EXIT - DATA FROM CUSTOMER SCREEN (DUMMY 1212)'. There are icons for adding and deleting entries on the right.
- Function codes:** A table with three columns: 'Program', 'Code', and 'Short Text'. It is currently empty, with icons for adding and deleting entries on the right.
- Dynpro Areas:** A table with six columns: 'Calling Dynpro', 'No.', 'Area', 'Called screen', 'No.', and 'Short Text'. It contains one entry: 'SAPLLMOB' (Calling Dynpro), '1212' (No.), 'EXIT0101' (Area), 'SAPLXLRF' (Called screen), '0000' (No.), and an empty 'Short Text' field. The 'Calling Dynpro' and 'No.' columns are highlighted with a green box, and the 'Called screen' column is highlighted with a red box. There are icons for adding and deleting entries on the right.

To create an alternative screen we need to copy screen **1212** from **SAPLLMOB** to **SAPLXLRF** with screen number **9212**. Below you can see attributes and code of 9212.

Batch field is assigned to LTAP-CHARG, which is WM transfer order's batch.

GS\_9212-CHARG is the verification field. GS\_9212 is a structure which is defined in ZXLRF TOP.

ZXLRF TOP is an include file of XLRF function group's master program LXLRF TOP.

Code of ZXLRF TOP:

```

*&-----
*& Include          ZXLRF TOP
*&-----

INCLUDE llmobdat.
INCLUDE zzxlr_f_9212.



DATA: gs_9212 TYPE ZWM_S_SCR_9212.

```

And attributes of structure ZWM\_S\_SCR\_9212

Dictionary: Display Structure

</





Screen number  Active

Attributes Element list Flow logic

General Texts/I/O Templates Special Display Mod. Groups/Functions References

H...	M	Name	Type ...	Line	Col...	De...	Vis...	Hei...	Scr...	Format	In...	Ou...	Out...	Dic...	Dict...	P
		<u>RLMOB-PENTER</u>	Push	19	1	9	9	1			<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input checked="" type="checkbox"/>	3	
		<u>RLMOB-PPGUP</u>	Push	19	11	3	3	1			<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input checked="" type="checkbox"/>	1	
		<u>RLMOB-PPGDN</u>	Push	19	15	3	3	1			<input type="checkbox"/>	<input type="checkbox"/>	<input type="checkbox"/>	<input checked="" type="checkbox"/>	1	
			OK	0	0	20	20	1	<input type="checkbox"/>	OK				<input type="checkbox"/>		

empty OK\_CODE

Screen number  Active

Attributes Element list Flow logic

Short Description Source Information - Multiple Materials

Original Language  English Package

Last Changed

Last Generation

Dynpro Type Settings

☐ Normal ☒ Subscreen ☐ Modal dialog box ☐ Selection Dynpro

☒ Switch Off R ☐ Template (N ☐ Hold Scroll F ☐ Without App ☐ Fiori: Zoom-

Other Attributes

Next Dynpro

Cursor Position

Screen Group

Rows/Columns Occupied

And the screen's code... Placed in include ZZXLRF\_9212 of Function Group XLRF

```

*&-----
*& Include          ZZXLRF_9212
*&-----

*&-----
*& Module STATUS_SCREEN_GROUP3 OUTPUT
*&-----
*&
*&-----
MODULE status_screen_group3 OUTPUT.

ENDMODULE.

*&-----
*& Module TAP_DISPLAY OUTPUT
*&-----
*&
*&-----
MODULE tap_display OUTPUT.

"Standard verification requires call of line below in sub screen.
PERFORM tap_display(sapllmob).

"Below lines are required for integration of batch verification
"and for enter behaviour.
"Standard enter behaviour does not know about added batch field GS_
"Therefore we need to create a mechanism to make sure it works as p
"Existing enter behaviour and verification
CONSTANTS: mem_9212(25) TYPE c VALUE 'GS_9212'.
FIELD-SYMBOLS: <scr9212> TYPE zwm_s_scr_9212.
ASSIGN (mem_9212) TO <scr9212>.

CONSTANTS: mem_R9212(25) TYPE c VALUE '(SAPLLMOB)GS_9212'.
FIELD-SYMBOLS: <gs_R9212> TYPE zwm_s_scr_9212.
ASSIGN (mem_R9212) TO <gs_R9212>.

IF <scr9212> IS ASSIGNED.

    IF <scr9212>-tapos IS INITIAL.
        <scr9212>-tapos = ltap-tapos.
    ELSE.
        "If screen is moved to next item
        IF <scr9212>-tapos NE ltap-tapos.

```

```
CLEAR <scr9212>.
<scr9212>-tapos = ltap-tapos.
IF <gs_R9212> IS ASSIGNED.
    CLEAR <gs_R9212>.
ENDIF.
ENDIF.
ENDIF.

LOOP AT SCREEN.

    IF screen-name EQ 'GS_9212-CCHARG'.
        IF ltap-charg IS INITIAL.
            screen-invisible = '1'.
            Screen-active = 0.

            IF <scr9212> IS ASSIGNED.
                CLEAR <scr9212>-mustverify.
            ENDIF.
        ELSE.

            IF <scr9212> IS ASSIGNED.
                <scr9212>-mustverify = abap_true.
            ENDIF.

            screen-invisible = '0'.
            Screen-active = 1.

            IF <scr9212> IS ASSIGNED AND ltap-charg EQ <scr9212>-cchang
                screen-input = 0.
            ELSE.
                screen-input = 1.
            ENDIF.

        ENDIF.
        MODIFY SCREEN.

    ENDIF.
ENDLOOP.

"This data is used in FIND_NEXT_KEY function
"Which is used in enter behaviour
IF <scr9212> IS ASSIGNED.
    IF <gs_R9212> IS ASSIGNED.
```

```

    <gs_R9212>-ccharg = <scr9212>-ccharg.
    <gs_R9212>-tapos = <scr9212>-tapos.
    <gs_R9212>-num_of_alq = <scr9212>-num_of_alq.
    <gs_R9212>-mustverify = <scr9212>-mustverify.
    <scr9212>-canverify = <gs_R9212>-canverify.
  ENDIF.
ENDIF.
ENDIF.
ENDMODULE.
*&-----
*& Module SPECIAL_FIELD_OUTPUT OUTPUT
*&-----
*&
*&-----
MODULE special_field_output OUTPUT.
ENDMODULE.
*&-----
*& Module SET_VALUES OUTPUT
*&-----
*&
*&-----
MODULE set_values OUTPUT.
ENDMODULE.
*&-----
*& Module SET_CURSOR OUTPUT
*&-----
*&
*&-----
MODULE set_cursor OUTPUT.
  "Required to set cursor order
  PERFORM set_cursor_pos(sapllmob)
    USING line_no_step_loop CHANGING screen_fcode current_field.
  PERFORM set_scroll(sapllmob) USING tap_index tap_init_size tap_total
    CHANGING temporary_index.
* Not for HUM
  PERFORM set_verify_fields_single(sapllmob).
  IF current_field IS INITIAL.
    LOOP AT SCREEN.
      IF screen-invisible = option_off
        AND screen-input = option_on
        AND screen-group1 <> 1.
        SET CURSOR FIELD screen-name.
      EXIT.
    ENDIF.
  ENDIF.

```

```

        ENDIF.
    ENDLOOP.
ENDIF.

ENDMODULE.

*&-----
*&      Module  EXIT_COMMANDS  INPUT
*&-----
*      text
*-----

MODULE exit_commands INPUT.

    PERFORM exit_commands_form(sapllmob)
        CHANGING ok_code screen_fcode prev_screen int_table_na

ENDMODULE.

*&-----
*&      Module  MODIFY_TAP    INPUT
*&-----
*      text
*-----

MODULE modify_tap INPUT.

    "Required to pass enhanced screen data to main program
    CONSTANTS: lv_memory(25) TYPE c VALUE '(sapllmob)rlmob'.
    FIELD-SYMBOLS: <rlmob> TYPE any.
    ASSIGN (lv_memory) TO <rlmob>.
    IF <rlmob> IS ASSIGNED.
        <rlmob> = rlmob.
    ENDIF.

ENDMODULE.

*&-----
*&      Module  USER_COMMANDS  INPUT
*&-----
*      text
*-----

MODULE user_commands INPUT.
ENDMODULE.

```

I have placed a piece data block in SAPLLMOB. That is because, we will need access

screens GS\_9212 data from other parts of call stack and SAPLLMOB sits on top of call stack. Go to include **LLMOBTOP** and add an enhancement at the end of include.

Code in **LLMOBTOP**;

```

*$$$-Start: (1)-----
ENHANCEMENT 2  ZLLMOBF01.      "active version
"For Enhacement screen used in LM06
data: GS_9212 TYPE zwm_s_scr_9212.
ENDENHANCEMENT.

```

And below code includes of enhancement MWMRF212.

EXIT\_SAPLLMOB\_170 -> EXIT\_SAPLLMOB\_170 -> we are passing values to sub screen.

```

*&-----
*& Include          ZXLRFU12
*&-----

"Screen exit for 9212 CMOD:ZWM -> Enhancement :MWMRF212 for RF Menu L
ltap          = i_ltap.
ltak          = i_ltak.
rlmob         = i_rlmob.
rl02b         = i_rl02b.
tap_init_size = i_tap_init_size.
tap_total_lines = i_tap_total_lines.
tap_index     = i_tap_index.
temporary_index = i_temporary_index.
line_no_step_loop = i_line_no_step_loop.
screen_fcode  = i_screen_fcode.
current_field = i_current_field.

```

EXIT\_SAPLLMOB\_480 -> INCLUDE ZXLRFU43 -> We are checking if Batch is verified. If not verified we will add an error line to make sure user gets a verification warning.

```
*&-----
*& Include          ZXLRFU43
*&-----
"Screen exit for 9212 CMOD:ZWM -> Enhancement :MWMRF212 for RF Menu L
"Verify batch
REFRESH o_verification_errors[].
IF gs_9212-canverify IS NOT INITIAL AND ltap-charg IS NOT INITIAL
  AND ( gs_9212-ccharg IS INITIAL OR gs_9212-ccharg NE ltap-charg ).

  CLEAR gs_9212-ccharg.

  DATA(error) = VALUE sprot_u( var1 = 'Batch' ) .
  APPEND error TO o_verification_errors.
ENDIF.
```

Enhancement code in function FIND\_NEXT\_KEY, add on top of existing code. We will prevent passing to next screen if batch is not verified!

```

FUNCTION FIND_NEXT_KEY.
.....
*$$$-Start: (1)-----
ENHANCEMENT 1  ZWM_ENTER_KEY_ROUTES.      "active version
"LM06 Screen has batch field and that field needs to be filled if TO
"This code is added to integrate custom batch field to enter navigati
  CONSTANTS: mem_R9212(25) TYPE c VALUE '(SAPLLMOB)GS_9212'.
  FIELD-SYMBOLS: <gs_R9212> TYPE zwm_s_scr_9212.
  ASSIGN (mem_R9212) TO <gs_R9212>.
  IF <gs_R9212> IS ASSIGNED.

      IF I_LAST_VERIF is NOT INITIAL or I_LAST_ITEM is NOT INITIAL.
          <gs_R9212>-canverify = abap_true.
      else.
          <gs_R9212>-canverify = abap_false.
      ENDIF.

      IF <gs_R9212>-mustverify is NOT INITIAL and <gs_R9212>-cchang i
          clear : I_LAST_VERIF ,I_LAST_ITEM.
      ENDIF.
  ENDIF.


ENDENHANCEMENT.

```

That is all you need to do to add a new field and integrate it with WM verification and enter behavior.

**Add new screen with a variant and change your LRFMD settings.**





Set screen with variant.

**Display View "Screen Management": Overview of Selected Set**

Mod. pool	ScrnFormat	Variant	Logi...	LogicalScr	Act.s...
SAPLLMOB	Narrow format	2	0212	Source info material	9212
SAPLLMOB	ITS format	2	0212	Source info material	9212
SAPLLMOB	Large format	2	0212	Source info material	9212

Change LRFMD Screen variant.

<b>Change View "Assignment User -&gt;Queue": Overview of Selected Set</b>								
New Entries								
Assignment User ->Queue								
W...	User	Queue	Status	Screen format	Variant	Main menu	Do	
TST	MEYRALIB		<input checked="" type="checkbox"/>	16X20	2	MAIN		

## Lets run LM06

I will fill first 2 verification fields and press enter without filling batch.

### Select Delivery

F2 Clr

F3 Back

F4 Nxt

Delivery

0080000051

### Source Information - Multiple

F1 Save

F2 Clr

F3 Back

F4 Nxt

Source information

Source Bin

SL1 01-01-02 ... L

Material information

Material

50

TL\_WHM - FERT

Batch 2000000152

1 PC

No.Aliquots N/A

S S

F5 Det

F6 Diff

Enter

### Message

Verification error in fields Batch


OK

That is it. I hope that post will help somebody who is also looking for that information.

Feel free to add your own solutions and suggestions to comments.

Thanks for reading.

## Comments



DominikTylczyn Active Contributor

2022 Aug 30  
3:31 PM

Nice one!!!




beyhan\_meyrali Active Contributor

2022 Aug 31  
7:09 AM

Thanks Dominik.

A comment like that from such an expert as yourself is much appreciated!

Regards



former\_member538062 Explorer

2023 Mar 14  
6:42 AM

Dear beyhan.meyrali

This is very useful document for activating batch verification, I'm surprised that SAP has not given this basic functionality with standard settings. Anyhow your article is very useful to achieve business requirement.

Regards,

Nayeem.



beyhan\_meyrali Active Contributor

2023 Mar 15

7:02 AM

Hi Nayeem,

Happy to see it is useful for you.

Regards

Powered by  
**Khoros** 